

# CASO PRÁCTICO PRÉSTAMOS. CÁLCULO DE PERIODOS Y VA

---

En este CASO práctico vamos a manejar otras funciones financieras para el cálculo de préstamos.

## CASO PRÉSTAMOS. CÁLCULO DE PERIODOS

---

Nuestra empresa necesita solicitar un préstamo de 60.100,00 € para adquirir una nueva furgoneta. El banco nos concede un préstamo de consumo aplicándonos un tipo de interés del 7%. El préstamo se amortizará según el tipo francés, es decir, con pagos constantes, teniendo la posibilidad de elegir el número de años en que vamos a amortizarlo.

### TRABAJO A REALIZAR:

1.- Sabiendo que podemos pagar una cuota mensual de 1.200,00 € determinar el **número de cuotas** que habrá que pagar para amortizar totalmente el préstamo.

2.- Dado que el número de periodos no resulta un número entero y que al pasarlo a años ocurre lo mismo, calcular la **cuota** exacta que deberíamos pagar para amortizar el préstamo mediante cuotas mensuales constantes a lo largo de toda la vida del préstamo aproximando el número de años al más cercano al resultado.

## CASO PRÉSTAMOS. CÁLCULO DEL VA

---

En esta ocasión queremos adquirir un local comercial en el centro de la ciudad. Para financiar dicha compra necesitaremos un préstamo. Tras consultar con varias entidades financieras, las mejores condiciones que hemos conseguido consisten en la concesión de un préstamo hipotecario por el importe total del local, a amortizar en 15 años mediante pagos trimestrales y a un tipo de interés del 2,5% nominal anual.

El director financiero nos indica que podemos disponer de 6.000,00 € trimestrales para pagar dicho préstamo.

## **TRABAJO A REALIZAR:**

- 1.- Calcular el importe máximo que podemos pagar por la compra del local.
  
- 2.- Supongamos ahora que podemos modificar el periodo de amortización del préstamo entre 10 y 20 años. Realizar un análisis de sensibilidad para determinar los importes que podríamos pagar por el local en función del número de años de amortización teniendo en cuenta que nuestras posibilidades de pago trimestral no varían.
  
- 3.- Realizar un nuevo análisis de sensibilidad considerando la variación anterior en el número de años y teniendo también en cuenta la posibilidad de que, simultáneamente, el tipo de interés oscile entre el 2 % y el 3 % en intervalos de un cuarto de punto porcentual.
  
- 4.- Realizar los cambios oportunos en la TABLA anterior para que se muestren únicamente los valores de VA en el caso de que éstos no superen los 320.000,00 € -límite que no deseamos sobrepasar en la compra del local- en caso de que se supere dicho límite, las celdas correspondientes deberán aparecer en blanco.

# SOLUCIÓN CASO PRÉSTAMOS. CÁLCULO DE PERIODOS

1.- Para resolver este CASO utilizaremos la función NPER:

**NPER:**

**=NPER(Tasa;Pago;Va;Vf;Tipo)**

Esta función calcula el número de periodos necesarios para amortizar un préstamo. Dado que en nuestro caso vamos a realizar pagos mensuales, la función NPER calculará el número de meses que tardaremos en amortizar el préstamo y para calcular los años multiplicaremos el resultado por la periodicidad de la operación, en este caso 12. No es lo mismo que calcular NPER con datos anuales ya que ambas operaciones no son “financieramente equivalentes”.

Por tanto, construiremos un modelo como este:


	A	B	C
1			
2			
3		<b>Importe del préstamo (VA)</b>	<b>60100</b>
4		<b>Tipo de interés anual</b>	<b>0,07</b>
5		interés del periodo (TASA)	=C4/C7
6		<b>Cuota mensual (PAGO)</b>	<b>1200</b>
7		<i>PERIODICIDAD 12</i>	
8			
9		<b>NUMERO DE MESES</b>	=NPER(C5;C6;-C3)
10			
11		<b>NUMERO DE AÑOS</b>	=C9/C7
12			
13			
14			

El resultado será el siguiente:


	A	B	C
1			
2			
3		<b>Importe del préstamo (VA)</b>	<b>60.100,00 €</b>
4		<b>Tipo de interés anual</b>	<b>7,00%</b>
5		interés del periodo (TASA)	0,58%
6		<b>Cuota mensual (PAGO)</b>	<b>1.200,00 €</b>
7		<i>PERIODICIDAD</i>	<i>12</i>
8			
9		<b>NUMERO DE MESES</b>	<b>59,4057954</b>
10			
11		<b>NUMERO DE AÑOS</b>	<b>4,95048295</b>
12			
13			

2.- Como vemos, el resultado del apartado anterior nos proporciona un valor que no es un número de años entero, por tanto, aproximaremos el resultado al número entero más próximo y con dicho dato calcularemos la nueva cuota que tendríamos que pagar para amortizar el préstamo.

Con el fin de que el modelo pueda funcionar ante un cambio en los datos intentaremos automatizar el proceso utilizando una función que aproxime el valor entero más próximo. Buscamos entre las funciones matemáticas y encontramos la función REDONDEAR:



Al pulsar “Aceptar” nos aparece el siguiente paso del asistente en el que indicaremos la celda que contiene el número de años y, como segundo argumento, indicaremos que redondee sin decimales:



Al pulsar de nuevo “Aceptar” la función quedará así:

=REDONDEAR(C11;0)

Debemos plantear un modelo que calcule la cuota a pagar en estas nuevas condiciones:

D	E	F
	<b>Importe del préstamo (VA)</b>	<b>=C3</b>
	<b>Tipo de interés anual</b>	<b>=C4</b>
	interés del periodo (TASA)	=F4/C7
	Número de años	=REDONDEAR(C11;0)
	Número de meses (NPER)	=F6*C7
	<b>PAGO MENSUAL</b>	<b>=PAGO(F5;F7;-F3)</b>

El resultado será:

	E	F
<b>Importe del préstamo (VA)</b>		<b>60.100,00 €</b>
<b>Tipo de interés anual</b>		<b>7,00%</b>
interés del periodo (TASA)		0,58%
Número de años		5
Número de meses (NPER)		60
<b>PAGO MENSUAL</b>		<b>1.190,05 €</b>

Podemos comprobar que, al redondear el número de años hasta el número entero más próximo, el préstamo se amortizará en 5 años y la cuota a pagar pasará a ser de 1.190,05 € en lugar del importe inicialmente previsto.

## SOLUCIÓN CASO PRÉSTAMOS. CÁLCULO DE VA

1.- La función que vamos a utilizar aquí también forma parte del grupo de funciones que hemos estudiado con anterioridad. En esta ocasión la función será:

**VA:**

$$=VA(Tasa;Nper;Pago;Vf;Tipo)$$

Esta función calcula el Valor Actual del préstamo, es decir, el importe del capital concedido en función del tipo de interés, el número de periodos y la cuota a pagar en cada periodo.

Al igual que en las otras funciones estudiadas, **Tasa**, **Nper** y **Pago** deben hacer referencia a un mismo periodo de tiempo, en este caso deseamos hacer pagos trimestrales por lo que dichos argumentos deberán referirse a periodicidad trimestral.

El modelo construido será:

	A	B	C
1			
2			
3		<b>Tipo de interés</b>	<b>0,025</b>
4		Tasa trimestral (TASA)	=C3/C8
5		<b>Periodo de amortización (en años)</b>	<b>15</b>
6		Número de trimestres (NPER)	=C5*C8
7		<b>Cuota trimestral (PAGO)</b>	<b>6000</b>
8		<b>PERIODICIDAD 4</b>	
9			
10		<b>VALOR DE COMPRA (VA)</b>	=-VA(C4;C6;C7)
11			
12			

El resultado en valores:

	A	B	C
1			
2			
3		<b>Tipo de interés</b>	<b>2,50%</b>
4		Tasa trimestral (TASA)	0,63%
5		<b>Periodo de amortización (en años)</b>	<b>15</b>
6		Número de trimestres (NPER)	60
7		<b>Cuota trimestral (PAGO)</b>	<b>6.000,00 €</b>
8		<b>PERIODICIDAD</b>	<b>4</b>
9			
10		<b>VALOR DE COMPRA (VA)</b>	<b>299.431,85 €</b>
11			

Es decir, si aceptamos el préstamo en las condiciones indicadas el precio que podemos pagar por el local será de 299.431,85 €

En los próximos apartados, vamos a realizar sendos análisis de sensibilidad para calcular los distintos importes que podríamos pagar por el local que queremos adquirir, ante distintos valores de algunos de los argumentos que intervienen en su función.

### **Análisis de Sensibilidad**

En ocasiones queremos conocer los distintos valores que tomaría una función al cambiar un dato por un rango de datos. Para ello podemos ir probando diferentes valores en la celda que contiene el dato que va a cambiar e ir anotando el resultado para cada valor. Esto se conoce como análisis de sensibilidad y Excel nos proporciona una herramienta para realizar dicho trabajo: la herramienta TABLA que se encuentra en la opción de menú:

DATOS/HERRAMIENTAS DE DATOS/ANÁLISIS Y SI/TABLA DE DATOS.

La TABLA puede ser de una variable, en caso de que solamente sea un dato el que vaya cambiando, o de dos variables cuando son dos los datos que queremos cambiar en la función que estamos calculando.

En el caso de una sola variable podemos colocar los datos en fila o en columna; en el caso de datos en fila, la función que deseamos calcular debe situarse en la celda inferior izquierda respecto a los datos:

	<b>Datos</b>
<b>Función</b>	

En caso de colocar los datos en columna, la celda que contiene la función debe ser la celda superior derecha respecto a los datos:

	<b>Función</b>
<b>Datos</b>	

En el caso de una TABLA de dos variables, la función se situará en la celda de intersección entre la columna de datos y la fila de datos.


<b>Función</b>	<b>Datos</b>
<b>Datos</b>	

2.- En este apartado deberemos plantear una TABLA de una variable, dado que intentamos conocer los distintos importes que tomaría el Valor Actual, es decir, el valor de compra del local, ante el cambio de un solo dato –en este caso, el número de años en que podemos amortizar el préstamo–.

En primer lugar situaremos en una columna los años a los que queremos referir el análisis. Teclearemos los dos primeros datos en las celdas B14 y B15 y mediante AUTOLLENADO completaremos la serie (también podemos utilizar la opción de menú INICIO/MODIFICAR/RELLENAR/SERIES).

El siguiente paso consistirá en seleccionar la celda en la que situaremos la fórmula; dado que hemos situado los datos en columna, elegiremos la celda de la fila superior y la columna derecha respecto a los datos –en nuestro caso C13-. En ella podríamos introducir de nuevo la función VA, sin embargo es preferible vincular esta celda con aquella en la que ya tenemos la función correspondiente, para ello, tecleamos el signo  $\square$  y a continuación hacemos clic con el ratón en la celda que contiene la función VA, en nuestro caso C10. De esta forma, Excel entiende que la celda C13 contiene lo mismo que la celda C10.


A continuación seleccionamos todas las celdas que formarán la tabla, es decir, el rango B13:C24, y vamos al menú DATOS/HERRAMIENTAS DE DATOS/ANÁLISIS Y SI/TABLA DE DATOS y aparecerá una pantalla como esta:



Tenemos que indicar en primer lugar cuál es la celda -de las contenidas en la fórmula o función- que va a ser sustituida en la misma por los valores que hemos situado en la fila. En nuestro caso no hay valores en fila, por tanto saltamos a la siguiente opción.

A continuación, tenemos que indicar cuál es la celda -de las contenidas en la fórmula o función- que va a ser sustituida en la misma por los valores que hemos situado en la columna. Recordamos que la función es:  $\square$  `=-VA(C4;C6;C7)` y los valores situados en la columna corresponden al número de años, que está representado en la función por la celda C5 (esta celda está contenida en la función VA a través de C6 que contiene en su fórmula a la celda C5).

Por tanto, para indicar la «Celda de entrada (columna)» haremos un clic con el ratón en la celda C5, con lo que la pantalla quedará así:




Al aceptar se calcula la TABLA automáticamente:

	A	B	C	D
12				
13			299.431,85 €	
14				
15			10 211.769,34 €	
16			11 230.186,45 €	
17			12 248.150,23 €	
18			13 265.671,84 €	
19			14 282.762,18 €	
20			15 299.431,85 €	
21			16 315.691,21 €	
22			17 331.550,36 €	
23			18 347.019,15 €	
24			19 362.107,18 €	
			20 376.823,84 €	

**3.-** La TABLA que debemos calcular en este apartado será de dos variables dado que pretendemos conocer los valores que tomaría el VA del préstamo al variar simultáneamente dos componentes de la función, el número de años y el tipo de interés.

Introduciremos en columna los datos relativos al número de años y en fila los distintos tipos de interés para los que deseamos realizar el análisis.

En la celda de intersección entre ambas series de datos –B27- introduciremos un vínculo a la celda que contiene la función VA, es decir, C10. Tras seleccionar todo el rango de celdas, que en nuestro caso será B27:G38, seleccionaremos la opción DATOS/TABLA y procederemos a introducir la referencia correspondiente, tanto en la “Celda de entrada (fila)” como en la “Celda de entrada (columna)”. De este modo el cuadro de diálogo quedará así:



Comprobamos de nuevo que las celdas a las que se hace referencia no aparecen directamente en la función VA, sin embargo, al igual que C5 estaba contenida en la fórmula de la celda C6, ocurre lo mismo con C3, el dato variable es el tipo de interés anual que está contenido en la función VA a través de la celda C4 en la que se calcula la tasa haciendo referencia al dato de C3.

Al aceptar, la TABLA se calcula automáticamente:

	A	B	C	D	E	F	G
26							
27		299.431,85 €	2,00%	2,25%	2,50%	2,75%	3,00%
28		10	217.033,37 €	214.378,88 €	211.769,34 €	209.203,88 €	206.681,63 €
29		11	236.449,39 €	233.288,69 €	230.186,45 €	227.141,42 €	224.152,38 €
30		12	255.481,91 €	251.778,94 €	248.150,23 €	244.594,03 €	241.108,69 €
31		13	274.138,48 €	269.858,96 €	265.671,84 €	261.574,84 €	257.565,71 €
32		14	292.426,54 €	287.537,82 €	282.762,18 €	278.096,59 €	273.538,14 €
33		15	310.353,36 €	304.824,45 €	299.431,85 €	294.171,70 €	289.040,24 €
34		16	327.926,09 €	321.727,53 €	315.691,21 €	309.812,23 €	304.085,87 €
35		17	345.151,71 €	338.255,58 €	331.550,36 €	325.029,95 €	318.688,48 €
36		18	362.037,08 €	354.416,93 €	347.019,15 €	339.836,27 €	332.861,09 €
37		19	378.588,93 €	370.219,70 €	362.107,18 €	354.242,32 €	346.616,38 €
38		20	394.813,83 €	385.671,85 €	376.823,84 €	368.258,92 €	359.966,64 €
39							

4.- Debemos conseguir que la TABLA anterior muestre los valores de la función VA, solamente en el caso de que se cumpla una determinada condición. Está claro que la solución requiere la utilización de una función condicional. Pero ¿dónde?:

Si intentamos introducir la función dentro de la matriz de resultados, Excel nos muestra un mensaje de error, es decir, no podemos realizar cambios en la TABLA una vez que se ha calculado.

La TABLA nos muestra los resultados de la fórmula o función introducida en la celda de intersección, cambiando los datos por los que aparecen en la fila y la columna de datos, por tanto, si

lo que se calcula es precisamente esa celda, está claro que esa es la celda en la que se debe introducir cualquier función que deseemos que afecte a los resultados de la TABLA.

Por tanto, será en la misma celda B27 donde debemos sustituir la simple referencia a la celda que contiene la función VA [=C10] por una función que podría ser: [=SI(C10<320.000,00;C10;"")].

Debemos poner un límite, consistente en un valor numérico que discrimine el resultado de la función condicional, sin embargo, debemos procurar que las fórmulas o funciones hagan referencia a celdas y no contengan datos. Para ello, introduciremos el valor del límite en una celda cualquiera, por ejemplo D41, y en la función condicional haremos referencia a dicha celda.

Por tanto, la función que introduciremos en B27 será: [=SI(C10<D41;C10;"")].

La TABLA quedará:

	A	B	C	D	E	F	G
26							
27		299.431,85 €	2,00%	2,25%	2,50%	2,75%	3,00%
28		10	217.033,37 €	214.378,88 €	211.769,34 €	209.203,88 €	206.681,63 €
29		11	236.449,39 €	233.288,69 €	230.186,45 €	227.141,42 €	224.152,38 €
30		12	255.481,91 €	251.778,94 €	248.150,23 €	244.594,03 €	241.108,69 €
31		13	274.138,48 €	269.858,96 €	265.671,84 €	261.574,84 €	257.565,71 €
32		14	292.426,54 €	287.537,82 €	282.762,18 €	278.096,59 €	273.538,14 €
33		15	310.353,36 €	304.824,45 €	299.431,85 €	294.171,70 €	289.040,24 €
34		16			315.691,21 €	309.812,23 €	304.085,87 €
35		17					318.688,48 €
36		18					
37		19					
38		20					
39							
40							
41			Límite	320.000,00 €			
42							

De este modo, si queremos hacer pruebas con distintos límites sólo tendremos que cambiar el valor en la celda D41 sin necesidad de alterar las fórmulas o funciones del modelo.

